Есть веб-сервис, пишу вызов его метода

```1c
    Определение = Новый WSОпределения(СокрЛП(Объект.АдресТест), СокрЛП(Объект.ЛогинТест), СокрЛП(Объект.ПарольТест));
    Прокси = Новый WSПрокси(Определение, "http://www.ponyexpress.ru/IntegratedExchange", "IntegratedExchange", "IntegratedExchangeSoap");

    //ТипWSПараметра = Прокси.ФабрикаXDTO.Пакеты.Получить("http://www.ponyexpress.ru/IntegratedExchange").Получить("SetGoods");
    ТипWSПараметра = Прокси.ФабрикаXDTO.Пакеты.Получить("http://www.ponyexpress.ru/IntegratedExchange").Получить("Goods");

    //Создаем параметр на основе типа
    WSПараметр = Прокси.ФабрикаXDTO.Создать(ТипWSПараметра);

    ЗаписьХМЛ = ПолучитьЗаписьXMLТовары();
    WSПараметр.Request = ЗаписьХМЛ;

    Ответ = Прокси.SetGoods();
```

В веб-сервисе есть такой метод SetGoods и у него есть параметр Request, тип параметра Goods (http://www.ponyexpress.ru/IntegratedExchange). См. на картинке.
Но! Вызов

```1c
    ТипWSПараметра = Прокси.ФабрикаXDTO.Пакеты.Получить("http://www.ponyexpress.ru/IntegratedExchange").Получить("Goods");
```

дает ошибку, нет такого типа. Этот вызов получает все, что в Пакете в ветке Свойства, а не Типы объектов.
А если я пишу так (SetGoods есть в списке свойств и здесь ошибки нет)

```1c
    ТипWSПараметра = Прокси.ФабрикаXDTO.Пакеты.Получить("http://www.ponyexpress.ru/IntegratedExchange").Получить("SetGoods");
```

то получаю ошибку дальше

```1c
    WSПараметр = Прокси.ФабрикаXDTO.Создать(ТипWSПараметра);
```

значение не может быть установлено свойству, имеющему объектный тип
я так понимаю, потому что тип параметра должен быть Goods, а я его сделал SetGoods.
Как все это правильно сделать?
---
ну, не XML, а XDTO. Это будет объект XDTO с типом Goods.
---
А так:

```1c
    ТипWSПараметра = Прокси.ФабрикаXDTO.Пакеты.Получить("http://www.ponyexpress.ru/IntegratedExchange").Получить("Goods").Тип;
```
---
хотя вру, наверно. Goods в какой ветке сидит? Типы значений?
---
да, Goods это в типах значений. Это тип параметра, который и надо установить.

```1c
    Прокси.ФабрикаXDTO.Пакеты.Получить("http://www.ponyexpress.ru/IntegratedExchange").Получить("Goods")
```

собственно и получает тип.
А SetGoods это метод веб-сервиса.
---
Вроде бы правильно должно быть так

```1c
    Определение = Новый WSОпределения(СокрЛП(Объект.АдресТест), СокрЛП(Объект.ЛогинТест), СокрЛП(Объект.ПарольТест));
    Прокси = Новый WSПрокси(Определение, "http://www.ponyexpress.ru/IntegratedExchange", "IntegratedExchange", "IntegratedExchangeSoap");

    ТипWSПараметра = Прокси.ФабрикаXDTO.Пакеты.Получить("http://www.ponyexpress.ru/IntegratedExchange").Получить("SetGoods");

    WSПараметр = Прокси.ФабрикаXDTO.Создать(ТипWSПараметра);

    ЗаписьХМЛ = ПолучитьЗаписьXMLТовары();
    WSПараметр.Request = ЗаписьХМЛ;

    Ответ = Прокси.SetGoods();
```

но получаю ошибку на строке WSПараметр.Request = ЗаписьХМЛ
Ошибка при установке значения атрибута контекста (Request)
 WSПараметр.Request = ЗаписьХМЛ;
по причине:
Несоответствие типов XDTO:
Тип '{http://www.ponyexpress.ru/IntegratedExchange}Goods' не найден
Значение не может быть установлено свойству, имеющему объектный тип
Ей надо как-то сказать, что мой XML имеет тип Goods из ее пространства имен, так, что ли? А как?
У меня этот XML имеет тип строка... начинается так
<?xml version="1.0" encoding="utf-8"?>
<d1p1:Goods xmlns:d1p1="Good" xmlns:xsi="http://www.ponyexpress.ru/IntegratedExchange">
---
А Request это не список? Если список, тогда надо
```1c
    WSПараметр.Request.Добавить(XMLТовар)
```
---
список товаров. А тогда XMLТовар это что будет? Кусок XML с описанием одного товара?
---
ну, не XML, а XDTO. Это будет объект XDTO с типом Goods.
---
спасибо большое! Направил в нужную сторону.
В итоге выглядит так

```1c
    Определение = Новый WSОпределения(СокрЛП(Объект.АдресТест), СокрЛП(Объект.ЛогинТест), СокрЛП(Объект.ПарольТест));
    Прокси = Новый WSПрокси(Определение, "http://www.ponyexpress.ru/IntegratedExchange", "IntegratedExchange", "IntegratedExchangeSoap");

    ТипОбъектаТовары = Прокси.ФабрикаXDTO.Тип("http://www.ponyexpress.ru/IntegratedExchange", "Goods");
    Товары = Прокси.ФабрикаXDTO.Создать(ТипОбъектаТовары);

    ТипОбъектаТовар = Прокси.ФабрикаXDTO.Тип("http://www.ponyexpress.ru/IntegratedExchange", "GoodItem");
    Товар = Прокси.ФабрикаXDTO.Создать(ТипОбъектаТовар);

    Товар.ID = "34343";

    Товары.Good.Добавить(Товар);

    Ответ = Прокси.SetGoods(Товары);
```

ну понятно, что надо переделать заполнение реквизитов товара, не в хмл писать, а в свойства объекта XDTO.